


Prof. Dr.-Ing. L. Palotas

FG Elektrische Übertragungstechnik
und Netzwerktheorie

FH Wiesbaden * FB 03 Elektrotechnik
University of Applied Sciences



Prof. Dr.-Ing. L. Palotas

Spice Kompakt

**Kurze Einführung in die
Schaltungssimulation**

mit SPICE

Simulation Program with Integrated Circuit Emphasis

2001

1 Schaltungsberechnung mit Simulatoren

Im folgenden wird die Installation von drei auf **SPICE**

"Simulation Program with Integrated Circuit Emphasis"

basierten Simulatoren

- **LTSPICE** (SWCADIII) der Fa. Linear Technologie
- **Spice3f5** (Universität Berkeley, USA)
- **PSPiceV9.x**, Demoversion, (Cadence, Orcad)

kurz beschrieben. Die ersten zwei Simulatoren sind "*freeware*" Vollversionen, wobei die letzte Berkeley Spice3 Simulator implementiert ist. *Empfehlenswert* ist der Simulator **LTSpice** (scad3) von *Lin. Tech.*, da er einerseits auch einen einfachen Schaltungseditor (Schematic-Entry) enthält, andererseits ist weitgehend kompatibel zu PSPice (auf Netzlisten-Ebene).

Die am Fraunhofer-Institut für Mikroelektronische Schaltungen und Systeme implementierte **Spice3f5**-Version verfügt über keine graphische Schaltungseingabe, die Simulationsergebnisse lassen sich jedoch auch in EXCEL anzeigen.

Der Simulator **PSPiceV9.x_Lite** ist die Demoversion des professionellen Simulators **PSPice9.x** mit den *Einschränkungen*: max. 64 Knoten, max. 10 Transistoren. Trotz dieser Einschränkungen lassen fast alle im Studium (Elektrotechnik, Netzwerktheorie, Elektronik, Digitaltechnik, Automatisierungstechnik etc.) gestellten Aufgaben lösen, bzw. die Lösungen kontrollieren. Mit Hilfe des mitgelieferten **PCB** - Demoprogramms kann auch das **Layout** (gedruckte Schaltung) entworfen werden.

1.1 LTSPICE

Hinweise zur Installation des Simulators LTSPICE (SWCADIII) der Fa. Linear Technologie.

0) Dateien: **swcad.zip** (1.35MB), **swcadinst.zip** (1.36MB) und **lib_pal.zip** (554kB) von dem homepage "www.ite.fh-wiesbaden.de/~palotas/" herunterladen (weitere Hinweise zur **Download-Seite** in der Vorlesung!) oder **swcadiii.exe** (2.834kB) direkt download von www.linear-tech.com.

1) Die Files **swcad.zip** und **swcadinst.zip** ==> extract in ein temp. Verzeichnis (z.B.: LT), (oder **swcadiii.exe** ==> extract in ein temp. Verzeichnis)

2) Aus diesem Verzeichnis installieren (**setup.exe**) angeboten wird in der Regel C:\programme\LTC\SWCADIII

3) Als Option: (nicht unbedingt erforderlich) Um deutsche "DIN" Symbole zu erhalten, nach der Installation das Verzeichnis **c:\programme\ltc\swcadIII\lib** umbenennen (z.B. lib_It)

Die Datei **lib_pal.zip** ==> extract in **c:\programme\ltc\swcadIII**

Das erstellte Verzeichnis **lib_pal** umbenennen in **lib**.

4) Temp. Verzeichnis (LT) löschen.

5) Programm scad3.exe starten.

Einige Filetypen:

Schaltungseitor-File:	*.asc
Circuit - file (ASCII):	*.cir (oder *.sp oder *.net)
Graphische Ausgabe:	*.raw (binär oder ascii, konfigurierbar)
Netzliste aus *.asc :	*.net (kann in *.cir umbenannt werden)
Symbole:	*.asy

Bemerkung:

LTSPICE ist kompatibel zu SPICE2G6, SPICE3 Simulatoren

PSpice *.CIR Files, die bei "file based simulation" erstellt worden, können mit LTSPICE zu 98% simuliert werden.

(installiert: ca 6.8 MB)

1.2 SPICE3F5

SPICE3F5 (1.12MB)

(Quelle: Prof. Dr.-Ing. Holger Vogt, Fraunhofer-Institut für Mikroelektronische Schaltungen und Systeme)

Simulator: (keine graphische Eingabe)jedoch graphische Ausgabe auch für EXCEL, Control - file, (Installiert ca. 3.1MB); Installiert wird in **C:\Spice_Win** mit mehreren Unterverzeichnissen. Im Unterverzeichnis "Bin" werden mit der Datei spice.rc einige Voreinstellungen konfiguriert.

```

Beispiel für spice.rc
* Standard spice and nutmeg init file
set SPICE_LIB_DIR=C:\Spice_Win\LIB
set sourcepath=C:\spice_win\examples
alias exit quit
alias acct rusage all
set filetype=ascii
* set filetype=binary
* set xlllineararcs
set hcopydevtype=postscript
set units=degrees
set history=3500

```

1.3 PSpice

PSpiceV9.2, Demoversion, CD "EDA for Windows 5.4" anfordern, (Vollinstallation ca. 230 MB, nur Simulator + Schematics ca. 113MB), oder download PSpice Student V9.1 (**91pspstu.exe** ca. 27.2MB). Einschränkungen: max. 64 Knoten, max. 10 Transistoren.

1.4 Hinweise

Fragen: ==> palotas@ite.fh-wiesbaden.de

homepage: www.ite.fh-wiesbaden.de/~palotas/

(Weitere Informationen in der Vorlesung beachten!)

2 Anweisungen (Spice2G6)

(Pspice, LTSpice, Spice3 kompatibel)

Passive Elemente		Quellen	
R	Widerstand	V	Unabhängige Spannungsquelle
C	Kapazität	I	Unabhängige Stromquelle
L	Induktivität	G	Spannungsgesteuerte Stromquelle
K	Gekoppelte Spulen	F	Stromgesteuerte Stromquelle
T	Verlustfreie Leitung	H	Stromgesteuerte Spannungsquelle
X	Elementanweisung für Teilschaltung	E	Spannungsgesteuerte Spannungsquelle

Halbleiterelemente		Modelle und Teilschaltungen	
D	Diode	.SUBCKT	Teilschaltung-
Q	Bipolarer Transistor		(Makro-Modell)-Definitioan
J	Sperrschicht-FET	.MODEL	Modelldefinition
M	MOSFET		default = Ersatzwert

DC Analyse Steuerung		Sonstige Anweisungen	
.DC	Übertragungskennlinie	Titel * +	Name der Schaltung Kommentar Fortsetzung einer Anweisung
.OP	Arbeitspunktanalyse	.END .ENDS	Ende der Schaltung Ende der Teilschaltung
.TF	Gleichstrom-Kleinsignalanalyse	.OPTION .WIDTH	Optionen Ausgabebreite
.NODESET	Anfangswerte der Knotenspannung bei AP-Berechnung	.PRINT .PLOT	Ausgabe Tabelle Ausgabe Druckerplot
.SENS	Empfindlichkeits-Analyse	.TEMP	Temperaturanalyse

AC Analyse Steuerung		Transienten Analyse Steuerung	
.AC	Wechselstrom-Kleinsignalanalyse	.TRAN	Einschwinganalyse
.NOISE	Rauschanalyse	.FOUR	Fourieranalyse
.DISTO	Verzerrungsanalyse	.IC	Anfangswerte der Knotenspannungen für Einschwinganalyse

Signal Generatoren (für Transienten Analyse)			
PULSE	Periodische Pulsfunktion	PWL	Stückweise lineare Funktion
SIN	Sinus oder gedämpfter Sinus		

Das an der Universität Berkeley (USA) entwickelte Programm **SPICE**

"Simulation Program with Integrated Circuit Emphasis"

simuliert das elektrische Verhalten von Schaltungen auf der Basis der Schaltelemente. Das sind bei SPICE insbesondere

- passive, konzentrierte Elemente (**R,L,C**, gekoppelte Spulen, **K**)
- ideale Leitungen (**T**)
- unabhängige und gesteuerte Quellen (**E,F,G,H**)
- Halbleiterbauelemente (**D,Q,J,M**)

Das elektrische Verhalten von Schaltungen kann auf verschiedene Art und Weise untersucht und dargestellt werden. SPICE bietet die folgenden Möglichkeiten:

- Gleichstromanalyse **.DC** (nichtlinear bzw. linearisiert)
- Nichtlineare Einschwinganalyse **.TRAN**
- Kleinsignal-Wechselstromanalysen **.AC**
- Rauschanalysen **.NOISE**

2.1 Ablauf der Simulation

Die aus konzentrierten Elementen und evtl. Leitungsstücken bestehende Schaltung muß in einer für die Berechnung der elektrischen Größen geeigneten Form, d.h. durch ein mathematisches Modell dargestellt werden.

Das Schaltungsmodell besteht aus einem System von im allgemeinen nichtlinearen Gleichungen und Differentialgleichungen, durch das die unbekanntenen Spannungen und Ströme miteinander verknüpft werden. Grundlage zur Erstellung dieses Gleichungssystems sind die mathematischen Modelle der einzelnen Schaltelemente.

SPICE enthält für jede zulässige Elementart ein allgemeines **Modell**, das noch vom Anwender frei wählbare **Parameter** aufweist, und somit eine individuelle Anpassung an das gegebene Schaltelemente ermöglicht.

Lineare passive Elemente besitzen in SPICE einfache Modelle und die Parameterwerte (z.B. Widerstand, Kapazität, Induktivität) werden durch die sogenannte *Elementzeile* eingegeben. Jedes einzelne Element ist in der Eingabe durch eine Elementzeile vertreten, in der auch Informationen über die Verbindungen zu anderen Elementen enthalten sind. Eine genauere Modellierung realer, linearer und passiver Schaltelemente ist durch geeignete Zusammenschaltung idealer Elemente möglich und ist Sache des Anwenders.

Die *Eingabeform von Halbleitern* weicht von dem eben beschriebenen Verfahren ab, da Halbleiter desselben Typs in der Regel mehrfach in einer Schaltung auftreten. Es wäre mühsam und unnötig, die Vielzahl der Parameter in allen Elementenzeilen gleichartiger Halbleitertypen einzugeben. Zur zahlenmäßigen Spezifikation von Halbleiterelementen gibt es daher

"Modellzeilen".

Alle Elementezeilen ein- und desselben Halbleitertyps beziehen sich dann auf eine Modellzeile dieses Typs, d.h. mehrere Elementezeilen benützen gewöhnlich eine Modellzeile.

Das Programm bildet mit Hilfe von Modell- und Elementezeilen zu jedem Schaltelement ein individuelles, zahlenmäßig vollständig definiertes mathematisches Modell.

Mit der in den Elementezeilen steckenden Information über die Struktur der Schaltung können nun die *Kirchhoffschen Gleichungen* für die Gesamtschaltung aufgestellt werden. Diese verknüpfen Klemmenströme und -spannungen der einzelnen Schaltelemente miteinander und führen so zu einem Gleichungssystem, das die gesuchten elektrischen Eigenschaften der gegebenen Schaltung beschreibt.

Das Resultat stellt offenbar ein mathematisches Modell einer aus physikalischen Elementen zusammengesetzten elektrischen Schaltung dar. Die durch Verbindungsleitungen zwischen den (konzentrierten) Elementen hervorgerufenen Effekte werden nicht automatisch mit erfaßt. Sie können jedoch vom Anwender durch Einführung zusätzlicher konzentrierter Elemente oder idealer Leitungen weitgehend berücksichtigt werden.

2.2 Simulationsarten

2.2.1 DC-Analyse (.DC - Anweisung)

Bei der Gleichstromsimulation wird die Schaltung zunächst so modifiziert, daß Spulen durch Kurzschlüsse und Kondensatoren durch Leerläufe ersetzt werden. Grundsätzlich beginnt jede Gleichstromanalyse mit der Berechnung des Arbeitspunktes. Dies erfordert eine einfache Lösung eines linearen Gleichungssystems, bei linearen (passiven) Schaltungen, andernfalls eine iterative Lösung eines nichtlinearen Gleichungssystems.

Gleichstrom-Arbeitspunkt: (.OP - Anweisung)

Format: .OP

Bemerkung: Diese Anweisung löst die Berechnung des Gleichstrom-Arbeitspunktes aus. Hierbei wird jede Spule durch einen Kurzschluß, jeder Kondensator durch einen Leerlauf ersetzt. (Small Signal Bias Solution = SSBS, wird vor jeder .TF, .SENS, .AC berechnet)

Erfolgt die *Arbeitspunkteinstellung* mehrmals hintereinander bei schrittweise veränderten Werten einer Quellengröße, so erhält man *Übertragungskennlinien*. Wird das Kleinsignalverhalten der Schaltung gewünscht, ist im Falle einer nichtlinearen Schaltung eine Linearisierung im Arbeitspunkt erforderlich. Unter der Annahme kleiner Abweichungen der Ströme und Spannungen vom Arbeitspunkt können dann Gleichstrom-Übertragungsfunktionen (Kleinsignal) und Empfindlichkeiten gegenüber Parameteränderungen ermittelt werden.

Format: **.DC qname1 vstart1 vstop1 vinkr1**
 + [**qname2 vstart2 vstop2 vinkr2**]

Beispiel: .DC VIN 0 10 0.1
 .DC VCR 0 2 0.1 IB 0 50UA 10UA

Hiermit erfolgt die (gleichstrommäßige) Berechnung einer Ausgangsgröße als Funktion der schrittweise veränderten Quellengröße *qname* (Übertragungskennlinie). Dabei stellt *qname* den Namen der Spannungs- oder der Stromquelle, **vstart** und **vstop** die Anfangs- und Endwerte und **vinkr** die Schrittweite dar. Zusätzlich kann eine zweite Quelle als Parameter ebenfalls inkremental variiert werden, diese hat die Berechnung einer Kurvenschar zur Folge.

2.2.2 AC-Analyse (.AC - Anweisung)

Die Wechselstromanalyse setzt voraus, daß die Schaltung nur Sinusquellen einheitlicher Frequenz enthält. Die Modelle nichtlinearer Schaltelemente werden im zuerst berechneten Arbeitspunkt linearisiert, d.h. die Ergebnisse sind in der Regel nur bei Kleinsignalbetrieb sinnvoll.

Mit Hilfe der komplexen Wechselstromrechnung werden für die so linearisierte Schaltung Amplituden und Phasenwinkel der gesuchten Netzwerkgrößen bei den vorgeschriebenen Frequenzen berechnet.

Bemerkung: Im Zuge der AC-Analyse kann auch das Rauschen der Schaltung als Funktion der Frequenz ermittelt werden. Das Programm generiert automatisch im Arbeitspunkt der Schaltung die äquivalenten Rauschquellen, und der Beitrag der einzelnen Quellen wird in einem gewünschten Schaltungsknoten aufsummiert. Auch kleine nichtlineare Verzerrungen können im Rahmen der Wechselstromanalyse simuliert werden.

DEC
 Format: **.AC OCT NP fstart fstop**
 LIN

Beispiel: .AC DEC 30 1KHz 1MEGHZ

Die **.AC-Anweisung** veranlaßt die Kleinsignalanalyse mit stationären Sinusquellen. Vor der AC-Analyse wird automatisch der Arbeitspunkt berechnet, und für die nichtlinearen Schaltungsteile das lineare Kleinsignalmodell ermittelt. Mindestens eine AC-Quelle muß im Netzwerk vorhanden sein. Großsignalquellen werden gleich Null gesetzt.

Es kann nur **eine** der drei Optionen **DEC** (*Frequenzvariation auf der Basis von Dekaden*), **OCT** (*Frequenzvariation auf der Basis von Oktaven*) **oder LIN** (*Lineare Frequenzvariation*) ausgewählt werden. Sie selektieren die Art der Intervalle, über die die Anzahl der Punkte, **NP**, pro Dekade verteilt werden.

2.2.3 Analyse von Einschwingvorgängen (.TRAN - Anweisung)

Die Simulation von Einschwingvorgängen zeichnet sich durch folgende Merkmale aus:

- Uneingeschränkte Berücksichtigung der Nichtlinearität der Schaltelementemodelle
- Konstante Quellengröße bis zum Zeitnullpunkt
- Berechnung des Zeitverlaufs in kleinen Zeitschritten ausgehend vom jeweiligen Schaltungszustand.

Da die Simulation zum Zeitpunkt $t=0$ startet, muß der Schaltungszustand in diesem Zeitpunkt in irgendeiner Form vorgeschrieben sein. Die Berechnung für das nächste Zeitinkrement erfolgt in zwei Schritten

- Einführung von Ersatzschaltungen für die Energiespeicher, bestehend aus nichtlinearen Widerständen und unabhängigen Quellen. Der Wert des Zeitinkrements wird aufgrund einer Fehlerabschätzung gewählt.
- Lösung des entstehenden nichtlinearen Gleichungssystems ähnlich wie bei der Arbeitspunktberechnung.

Ein Teil des Ergebnisses kann anschließend einer Fourier-Analyse unterzogen werden.

(Im weiteren [] bedeutet: optionell)

Format: **.TRAN tstep tstop [tstart [tmax]][UIC]**

Beispiel:

```
.TRAN 50NS 1US UIC
.TRAN 50NS 10US 9US
```

Die **.TRAN** Anweisung veranlaßt die Transienten-Analyse. **tstep** ist dabei die Schrittweite für die Ausgabe und **tstop** der Endzeitpunkt, bis zu der die Analyse vollzogen werden soll. **tstart** gibt den ersten Zeitpunkt an, ab dem die **Daten ausgegeben** werden sollen. Ist dieser Wert nicht angegeben, so wird er auf Null gesetzt. **tmax** stellt die maximale rechnerinterne Schrittweite dar. Für die Analyse muß mindestens eine **Großsignalquelle** (Signalgenerator) im Netzwerk vorhanden sein.

Die UIC-Anweisung ist einzugeben, falls die bei L und C mit **IC=...** angegebenen Anfangswerte eingesetzt werden sollen. Folgende Tabelle zeigt den Einfluß der einstellbaren Anfangsbedingungen (UIC, .IC, IC) bei der Transientenanalyse:

UIC	.IC	IC	Art der Berücksichtigung der Anfangsbedingung (AB)
nein	nein	nein	ITS = SSBS
nein	nein	ja	ITS = SSBS (IC ohne Einfluß auf ITS)
nein	ja	nein	ITS mit .IC berechnet
nein	ja	ja	ITS mit .IC berechnet (IC ohne Einfluß auf ITS)
ja	nein	nein	kein ITS, alle AB=0
ja	nein	ja	kein ITS, AB = IC, (fehlende AB = 0)
ja	ja	nein	kein ITS, VAB = .IC, (fehlende VAB = 0, IAB = 0)
ja	ja	ja	kein ITS, AB = IC, (fehlende VAB = .IC, IAB = 0)

wobei ITS = Initial Transient Solution (DC Arbeitspunktanalyse)

SSBS = Small Signal Bias Solution (bei .OP)

.IC

Anfangsbedingung für Knotenspannungen

Format: `.IC V(N1)=val1 ... V(Nj)=valj`

Beispiel: `.IC V(3)=6.8V V(4)=1.25V V(5)=-3.12V`

Die **.IC**-Anweisung arbeitet in zwei verschiedenen Wegen, abhängig davon, ob das UIC-Kommando in der **.TRAN**-Anweisung vorhanden ist oder nicht. (siehe Tabelle nach der **.TRAN** Anweisung!)

2.3 Eingabe der Schaltung

Die zu simulierende Schaltung wird durch eine Reihe von **Elementzeilen** (Elementdefinitionen) beschrieben. Durch diese werden Art und Zahlenwerte der Schaltelemente und die Netzwerktopologie festgelegt. Zusätzlich müssen Steuerkommandos (Steuerzeilen) eingegeben werden; diese ermöglichen die Kontrolle des Programmlaufs. Die **erste Zeile** einer Eingabe muß eine **Titelzeile** sein, die letzte Zeile dient ausschließlich zur Markierung des Eingabeendes. Damit ergibt sich der folgende grundsätzliche Aufbau der Eingabe:

Titelzeile (muß die erste Anweisung sein!)

Elementzeilen,

Teilschaltung (Subcircuit)– Definitionen,

Steuerzeilen und Kommentarzeilen in beliebiger Reihenfolge


.END

Die Schaltungsknoten müssen mit ganzen, positiven Zahlen numeriert werden. Der Bezugsknoten erhält die Nummer 0. Jeder Knoten **muß** über einen **Gleichstrompfad** vom **Bezugsknoten** aus erreichbar sein. (Bei den neuen Spice3 Implementierungen können als Schaltungsknoten auch Buchstaben sein!)

Jedem Element der Schaltung wird eine **Elementzeile** zugeordnet, die den Namen des Elements enthält, weiterhin die Nummern der Knoten, mit denen das Element verbunden ist, und die Zahlenwerte zur quantitativen Charakterisierung der elektrischen Eigenschaften. Durch den **ersten Buchstaben** eines Elementnamens wird die **Art des Schaltelements** definiert. Maximal sieben zusätzliche alphanumerische Zeichen vervollständigen den Namen des Elements.

In eckige Klammern [...] gesetzte Datenfelder sind keine notwendigen Bestandteile der Elementzeilen und enthalten optionale Eingabedaten.

Zählpfeile für Strom und Spannung haben, auch bei Quellen, grundsätzlich **dieselbe Richtung**, nämlich von N_+ nach N_- !



2.4 Eingabeformat, Ausgabe

Die Eingabe ist **formatfrei**.

Trennzeichen für Felder sind: Leerzeichen, ein Komma, ein Gleichheitszeichen, eine rechte oder eine linke Klammer. Zusätzliche Zwischenräume werden überlesen.

Eine **Fortsetzungszeile** zur vorhergehenden Zeile muß mit dem Pluszeichen "+" beginnen.

Kommentarzeilen werden mit einem Stern "*" eingeleitet.

Ein **Namenfeld** beginnt mit einem Buchstaben und darf keine Begrenzungszeichen (Delimiter) enthalten.

Ein **Zahlenfeld** ist

eine Integerzahl (z.B. 3,-44)

oder eine Gleitkommazahl (z.B. 2.536,-0.43)

oder eine der genannten Zahlen mit zusätzlicher ganzzahliger Zehnerpotenz als Faktor (z.B. 2.536E3,-44E-8,-5.1E10).

Zehnerpotenzen können auch durch Abkürzungen (**Skalenfaktoren**) ersetzt werden, siehe folgende Tabelle:

Potenzdarstellung	E-15	E-12	E-9	E-6	E-3	E3	E6	E9
äquival. Skalenfaktor	F	P	N	U	M	K	MEG	G

Die **Skalenfaktoren** müssen *unmittelbar* an das Zahlenfeld anschließen. Weitere unmittelbar folgende Buchstaben werden ignoriert.

Hiermit sind übersichtliche Darstellungen von Zahlenwerten mit Einheiten möglich, z.B.

$$10\text{MEGOHM} = 1\text{E}7 = 10000000 = 10\text{MEG}$$

Schließen von den Skalenfaktoren abweichende Buchstaben unmittelbar an ein Zahlenfeld an, so werden diese ebenfalls ignoriert.

Das Programm geht grundsätzlich davon aus, daß physikalische Größen in **SI-Einheiten** eingegeben werden. Ein evtl. Anhängen der Einheiten an das Zahlenfeld dient lediglich der besseren Lesbarkeit der Eingabe.

Die Ergebnisausgabe als Tabelle:

Format: **.PRINT type var [var2 ... varn]**

Beispiel: `.PRINT AC V(1) VM(5,4) VP(5,4) IDB(VM2)`

Diese Anweisung beschreibt die Ausgabe der Zahlenwerte als Tabelle. Es können bis zu 8 Ausgangsvariablen ausgegeben werden. Der **type**-Parameter gibt die Analyseart an, deren Ergebnis auszugeben ist. Es muß einer aus der folgenden Liste sein: TRAN, AC, DC, NOISE oder DISTO. Ausgabevariablen beginnen mit den Buchstaben **V** oder **I** und stellen die Spannung oder den Strom dar. Der Strom kann nur über einer Spannungsquelle gemessen werden. *Um also Strom messen zu können, muß eine Spannungsquelle mit der Spannung Null eingeführt werden.*



Für die **AC-Analyse** können noch folgende Symbole an V oder I angehängt werden:

VR	bzw.	IR	Realteil
VI	bzw.	II	Imaginärteil
VM	bzw.	IM	Betrag
VP	bzw.	IP	Phase
VDB	bzw.	IDB	$20 \cdot \log(V_M/\text{Volt})$ bzw. $20 \cdot \log(I_M/\text{Ampere})$

2.5 Unzulässige Schaltungen


Aufgrund des Rechenverfahrens (modifizierte Knotenanalyse) können Schaltungen mit folgenden Eigenschaften nicht simuliert werden:

- Es dürfen keine Maschen auftreten, die nur aus Spannungsquellen und/oder Induktivitäten bestehen.



Beispiele:


- Es darf nicht möglich sein, in der Schaltung eine geschlossene Linie zu zeichnen, die nur von Zweigen mit Kapazitäten und/oder Stromquellen durchstossen wird. Beispiel:



2.6 Bauelemente

2.6.1 Passive Elemente

R Widerstand



Format: **Rname** N_1 N_2 **zahlenwert** [**TC** = t_{c1} [, t_{c2}]]


Beispiel: R1 1 2 1K
 RSOURCE 105 32 1MEGOHM TC =.001,.015

Der Name muß mit dem Buchstaben **R** beginnen. N_1 und N_2 sind die Knotennummern. Der Zahlenwert kann sowohl positiv als auch negativ sein, aber *nicht* Null. Optionale Daten werden bei nichteingabe auf Null gesetzt.

Temperaturabhängigkeit des Widerstandswertes:

$$R(T) = R(T_{nom}) \cdot (1 + t_{c1}(T - T_{nom}) + t_{c2}(T - T_{nom})^2)$$

Es gilt $T_{nom} = 27^\circ C$, falls nicht in der Option-Anweisung ein anderer Wert festgelegt wird.

C**Kapazität**

Format: **Cname** N_+ N_- [**POLY**] **zahlenwert** [**C1** [**C2** ...]] [**IC=v**]

Beispiel:

```
CLOAD 5 0 10UF
C2 3 2 POLY 1UF .1UF .05UF
C1 8 0 .01UF IC=10V
```


Der Name muß mit dem Buchstaben **C** beginnen. N_+ und N_- sind die Knotennummern. Die Polarität wird für die Anfangsbedingung und der nichtlinearen Polynomkoeffizienten benötigt. Bei Einsatz der **POLY**-Anweisung wird die Kapazität als nichtlineare Funktionen der Spannung betrachtet.

Er berechnet sich wie folgt:

$$C(V) = \text{zahlenwert} + C_1 \cdot V + C_2 \cdot V^2 + \dots + C_n \cdot V^n$$

mit $n \leq 20$

Die Anfangsbedingung für die Transientenanalyse wird mit **IC=u** eingestellt. Sie setzt die Anfangsspannung auf den Wert U . Hierzu muß die **UIC**-Anweisung in der **.TRAN** Anweisung stehen.

L**Induktivität**

Format: **Lname** N_+ N_- [**POLY**] **zahlenwert** [**L1** [**L2** ...]] [**IC=i**]

Beispiel:

```
L5 5 3 10HY
LSR 3 2 POLY .01 0 -.001 0 0.0001
L1 8 0 10MHY IC=10MA
```

Der Name muß mit dem Buchstaben **L** beginnen. N_+ und N_- sind die Knotennummern. Die Polarität wird für die Anfangsbedingung und der nichtlinearen Polynomkoeffizienten benötigt. Bei Einsatz der **POLY**-Anweisung wird die Induktivität als nichtlineare Funktion des Stromes betrachtet.


Er berechnet sich wie folgt:

$$C(V) = \text{zahlenwert} + L_1 \cdot I + L_2 \cdot I^2 + \dots + L_n \cdot I^n$$

mit $n \leq 20$

Die Anfangsbedingung für die Transientenanalyse wird mit **IC=i** eingestellt. Sie setzt den Anfangsstrom auf den Wert i . Hierzu muß die **UIC**-Anweisung in der **.TRAN** Anweisung stehen. Sättigungsbereiche können mit der Polynomerweiterung approximiert werden.

K Gekoppelte Spulen



Format: **Lname2 N₁ N₂ zahlenwert [IC=i₁(0)]**

Lname3 N₃ N₄ zahlenwert [IC=i₂(0)]

Kname1 Lname1 Lname2 zahlenwert des Kopplungsfaktors

Beispiel:

```
L1 3 5 10UHY
L6 2 0 1UHY
K12 L1 L6 .999
```

L1 und **L2** werden als normale Induktivitäten definiert. Die magnetische Kopplung wird durch eine weitere Eingabezeile festgelegt.


$$u_1 = L_1 \frac{dI_1}{dt} + M_{12} \frac{dI_2}{dt}$$

$$u_2 = M_{12} \frac{dI_1}{dt} + L_2 \frac{dI_2}{dt}$$

$$K = \frac{|M_{12}|}{\sqrt{L_1 \cdot L_2}} ; \quad (0 \leq k < 1)$$

Die Festlegung der Knoten **N₁**, **N₂**, **N₃** und **N₄** ist so zu wählen, daß die Gegeninduktivität **M₁₂** positiv ist.

T Verlustfreie Leitung Transmission Line



Format: **Tname N1 N2 N3 N4 ZO=zahlenwert1 TD=zahlenwert2**
+ [F=frequenz [NL=nlänge]]

Beispiel: **T1 1 0 2 0 ZO=50 TD=25NS**

Die Beschreibung der verlustfreien Spule erfolgt durch den Wellenwiderstand **Z₀** und der Laufzeit **T_D**. Die zusätzlichen Parameter für die Länge der Leitung können die Laufzeitspezifikation **T_D** ersetzen. **F** ist die Frequenz und **NL** ist die Zahl der Wellenlänge, welche sich über die Leitung bei dieser Frequenz fortsetzt. Wird **NL** nicht angegeben, so wird sie auf 0.25 gesetzt.

werden.

Der AC Parameter (Sinusquellen ohne Gleichanteil) wird für die stationäre, lineare Wechselstromanalyse benötigt. Der Betrag von AC wird für die Verstärkungsmessung auf eins gesetzt. Der Nullphasenwinkel muß im Gradmaß angegeben werden.

2.6.3 Signalgeneratoren

PULSE Periodische Pulsfunktion

Format: ... PULSE w1 w2 [td [tr [tf [pw [per]]]]]

	Parameter	Ersatzwerte	
w1	Initialisierungswert	Muß definiert werden	
w2	Pulswert	Muß definiert werden	
td	Verzögerungszeit	0	
tr	Anstiegszeit	TSTEP	
tf	Abfallzeit	TSTEP	
pw	Pulsweite	TSTOP	
per	Periodendauer	TSTOP	

Die Periodendauer beinhaltet nicht die Verzögerungszeit. ($w1 \neq w2$)

SIN Gedämpfte sinusfunktion mit Gleichanteil


Format: ... SIN w0 wa [freq [td [α]]]

	Parameter	Ersatzwerte	
w0	Offset	Muß definiert werden	
wa	Amplitude	Muß definiert werden	
freq	Frequenz	1 / TSTOP	
td	Verzögerungszeit	0	
α	Dämpfungskonstante	0	

$$w(t) = w0 + wa \cdot e^{-(t-t_d) \cdot \alpha} \cdot \sin[2 \cdot \pi \cdot freq \cdot (t - t_d)]$$

Die Amplitude kann exponentiell mit dem Dämpfungsfaktor α an- oder abklingen.

PWL Stückweise lineare Funktion




Format: **... PWL t1 w1 t2 w2 ... tn wn**

Die Funktionswerte zwischen zwei benachbarten Stützpunkten werden linear interpoliert.

2.6.4 Gesteuerte Quellen

(PSpice, LTSpice kompatibel, POLY ist in Spice3 nicht implementiert!)

E Spannungsgesteuerte Spannungsquelle (VCVS)



Format: **Ename N+ N- [POLY (nd)] NC1+ NC1- [NC2+ ...]
+ [P0 P1 ...] [IC=val1,val2,...]**

Beispiel: **E1 3 4 21 17 1
ESUM 3 0 POLY(3) 12 0 9 0 5 0 0 1 1 1**

Der Name muß mit dem Buchstaben **E** beginnen. N_+ und N_- sind die positiven und negativen Knotennummern. $NC1+$, $NC1-$, ... sind die positiven und negativen Anschlüsse der Steuerquelle. POLY(nd) wird für mehrdimensionale Steuerung der Spannungsquelle gebraucht. Wird dieser Wert nicht angegeben, so erfolgt die Standarteinstellung 1.


Die Spannung berechnet sich für eine Steuerquelle wie folgt:

$$f_1 = V(NC1+) - V(NC1-)$$

$$U_0 = P_0 + P_1 \cdot f_1 + P_2 \cdot f_1^2 + \dots + P_n \cdot f_1^n$$

Bemerkung: Wird nur ein Koeffizient angegeben, so wird dieser automatisch als **P1** angenommen und P0 auf 0 gesetzt.

G Spannungsgesteuerte Stromquelle (VCCS)



Format: **G**name N+ N- [**POLY** (nd)] NC1+ NC1- [NC2+ ...]
+ [**P0 P1 ...**] [**IC=val1,val2,...**]

Beispiel: `G1 2 3 5 0 10000UMHOS`
`GNL 3 0 POLY(3) 12 0 9 0 5 0 0 1M 1M 1M`

Der Name muß mit dem Buchstaben **G** beginnen. N_+ und N_- sind die positiven und negativen Knotennummern. **NC1+**, **NC1-**, ... sind die positiven und negativen Anschlüsse der Steuerquelle. **POLY(nd)** wird für mehrdimensionale Steuerung der Spannungsquelle gebraucht. Wird dieser Wert nicht angegeben, so erfolgt die Standarteinstellung **1**.


Die Spannung berechnet sich für eine Steuerquelle wie folgt:

$$f_1 = V(\text{NC1 } +) - V(\text{NC1 } -)$$

$$I_0 = P_0 + P_1 \cdot f_1 + P_2 \cdot f_1^2 + \dots + P_n \cdot f_1^n$$

Bemerkung: Wird nur ein Koeffizient angegeben, so wird dieser automatisch als **P1** angenommen und **P0** auf 0 gesetzt.

H Stromgesteuerte Spannungsquelle (CCVS)



Format: **H**name N+ N- [**POLY** (nd)] VN1 [VN2 ...]
+ [**P0 P1 ...**] [**IC=val1,val2,...**]

Beispiel: `H1 3 4 VCC 2MA 0.25`
`HAMP 3 0 POLY(3) V1 V2 V3 0 1 1K 1K`

Der Name muß mit dem Buchstaben **H** beginnen. N_+ und N_- sind die positiven und negativen Knotennummern. **VN1**, **VN2**, ... ist die Steuerquelle. **POLY(nd)** wird für mehrdimensionale Steuerung der Spannungsquelle gebraucht. Wird dieser Wert nicht angegeben, so erfolgt die Standarteinstellung **1**.

Die Spannung berechnet sich für eine Steuerquelle wie folgt:


$$f_1 = I(VN1)$$

$$U0 = P_0 + P_1 \cdot f_1 + P_2 \cdot f_1^2 + \dots + P_n \cdot f_1^n$$

Bemerkung: Wird nur ein Koeffizient angegeben, so wird dieser automatisch als P1 angenommen und P0 auf 0 gesetzt.

Der **steuernde** Strom **I** muß durch eine ideale *Spannungsquelle* fließen (zu definieren als unabhängige Quelle Vname). Eventuell muß diese Quelle **zusätzlich** in den Pfad des Stromes I (mit der Spannung 0V) eingeführt werden.

F Stromgesteuerte Stromquelle (CCCS)



Format: **F**name N+ N- [**POLY** (nd)] VN1 [VN2 ...]
+ [**P0 P1 ...**] [**IC=val1,val2,...**]

Beispiel: **F1** 3 4 VCC 2MA 0.25
FAMP 3 0 POLY(3) V1 V2 V3 0 1 1 1

Der Name muß mit dem Buchstaben **F** beginnen. N₊ und N₋ sind die positiven und negativen Knotennummern. **VN1**, **VN2**, ... ist die Steuerquelle. POLY(nd) wird für mehrdimensionale Steuerung der Spannungsquelle gebraucht. Wird dieser Wert nicht angegeben, so erfolgt die Standarteinstellung 1.

Die Spannung berechnet sich für eine Steuerquelle wie folgt:

$$f_1 = I(VN1)$$

$$I0 = P_0 + P_1 \cdot f_1 + P_2 \cdot f_1^2 + \dots + P_n \cdot f_1^n$$

Bemerkung: Wird nur ein Koeffizient angegeben, so wird dieser automatisch als P1 angenommen und P0 auf 0 gesetzt.


Der *steuernde* Strom I muß durch eine ideale Spannungsquelle fließen (zu definieren als unabhängige Quelle **Vname**). Eventuell muß diese Quelle zusätzlich in den Pfad des Stromes I (mit der Spannung 0V) eingeführt werden.

2.7 Teilschaltungen (Makromodelle, Subcircuits)

Format: **.SUBCKT** subname k1 [k2 [k3 ...]]

-
- **Elementdefinition zur Teilschaltung-Beschreibung**
- **(keine Steueranweisungen)**
-
- .ENDS** [subname]

Beispiel:



```
.SUBCKT 7400      1    11    9    6
*                IN1  IN2    OUT  VCC
*                C    B    E
Q1      3    2    1    TTLQ
Q2      3    2    11   TTLQ
Din1    0    1                TTLD
Din2    0    11           TTLD
Q3      4    3    5    TTLQ
Rb12    6    2                4K
RC3     6    4                1.6k
RC4     6    7                130
RE2     5    0                1K
Q4      7    4    8    TTLQ
Dout    8    9                TTLD
Q5      9    5    0    TTLQ
.MODEL  TTLD  D  (RS=40 TT=0.1NS CJO=0.9PF)
.MODEL  TTLQ  NPN (BF=150 RB=70 RC=40 CCS=2PF
+ TF=0.1NS TR=10NS CJE=9PF CJC=1.5PF VA=50)
.ENDS 7400
```

Die Teilschaltung-SUBCIRCUIT-Definition (**Makro-Modell-Definition**) beginnt mit **.SUBCKT** und muß mit **.ENDS** abgeschlossen werden. **Subname** ist der Name der Teilschaltung, mit dem die Teilschaltung vom Hauptprogramm aus **aufgerufen** wird. Die **Anschlußnummern** sind in der **gleichen Reihenfolge**, wie sie im **Teilschaltungsaufruf** (**X**-Anweisung) stehen, zugeordnet. Anschluß 0 kann nicht als Übergabeparameter definiert werden. Er ist in der Teilschaltungsbeschreibung der gleichen Nummer wie im Hauptteil zugeordnet. Bei **verschachtelten** Teilschaltungen folgt in der **.ENDS** - Anweisung der Name **subname** der aktuellen Teilschaltung.

X **Teilschaltungsaufruf** (Pseudo - Elementanweisung)

Format: **Xname** N1 [N2 [N3 ...]] **subname**

Beispiel: XU1 1 2 3 4 5 NAND

Der **Teilschaltungsaufruf** beginnt mit dem Namen **X**. Die **Reihenfolge** der Anschlußnummer erfolgt in der **gleichen Weise**, wie in der **Teilschaltungsbeschreibung**. Es können bis zu 20 Anschlußnummern übergeben werden. Die Teilschaltung besteht aus einer Gruppe von Elementen beliebiger Anzahl und Komplexität. Der Subname wird bei Ausführung des Programms durch diese Elemente ersetzt. Teilschaltungen können **verschachtelt** auftreten, jedoch **nicht rekursiv** (zirkular).

2.8 Nichtlineare mehrdimensionale Quellen

Format: $\dots \text{POLY}(\text{nd}) \dots \dots [\text{P}_0 \text{P}_1 \dots]$

$\text{POLY}(\text{nd})$: \mathbf{nd} - dimensionales Polynom

$\text{P}_0 \text{P}_1 \dots$: Polynomkoeffizienten

x_1, x_2, \dots, x_n Kontroll-Variablen (Spannung oder Strom)

x_o Ausgangsgröße (Spannung oder Strom)

Es gilt für ein \mathbf{n} - dimensionales Polynom

$$\begin{aligned}
 x_o(x_1, x_2, \dots, x_n) = & \\
 & P_0 + \\
 & P_1 \cdot x_1 + P_2 \cdot x_2 + \dots P_n \cdot x_n + \\
 & P_{n+1} \cdot x_1 \cdot x_1 + P_{n+2} \cdot x_1 \cdot x_2 + \dots P_{n+n} \cdot x_1 \cdot x_n + \\
 & P_{2n+1} \cdot x_2 \cdot x_2 + P_{2n+2} \cdot x_2 \cdot x_3 + \dots P_{2n+n-1} \cdot x_2 \cdot x_n + \\
 & \dots \\
 & P_{n!/(2(n-1)!)+2n} \cdot x_n \cdot x_n + \\
 & P_{n!/(2(n-2)!)+2n+1} \cdot x_1^2 \cdot x_1 + P_{n!/(2(n-2)!)+2n+2} \cdot x_1^2 \cdot x_2 + \dots \\
 & \dots
 \end{aligned}$$

Beispiel: für $n_d = 2$

$$\begin{aligned}
 x_o(x_1, x_2) = & P_0 + P_1 \cdot x_1 + P_2 \cdot x_2 + P_3 \cdot x_1^2 + P_4 \cdot x_1 \cdot x_2 + P_5 \cdot x_2^2 + \\
 & P_6 \cdot x_1^3 + P_7 \cdot x_1^2 \cdot x_2 + P_8 \cdot x_1 \cdot x_2^2 + P_9 \cdot x_2^3 + P_{10} \cdot x_1^4 + \dots
 \end{aligned}$$

3 Eingabe von Halbleiterbauelementen

Bei der Eingabe einer Schaltung bezieht sich jedes Halbleiterbauelement auf ein Halbleitermodell, welches durch eine Modelldefinition beschrieben wird.

Das Programm stellt Modelle in allgemeiner Form für folgende Halbleitertypen zur Verfügung

- Dioden (**D**)
- Sperrschicht-FET (**J**)
- Bipolar Transistoren (**Q**)
- MOSFET (**M**)

Die Modellzeilen zur Definition der Halbleitermodelle haben folgende Form

Format: **.MODEL** *modname typ* [*parameter1=wert1 parameter2=wert2 ...*]

Beispiel: `.MODEL MOD1 NPN(IS=0.1P BF=180 VAF=100)`
`.MODEL DGAS D(N=2)`

Die **.MODEL** Anweisung definiert eine Liste von Modellparameter, welche von einem oder mehreren Elementzeilen aus aufgerufen wird. **modname** ist der Name des Halbleitermodells aus der Elementzeile des entsprechenden Aufrufs. Für "**typ**" ist eines der folgenden Schlüsselwörter zu setzen:

NPN	Bipolarer npn-Transistor	NJF	n-Kanal Sperrschicht-FET
PNP	Bipolarer pnp-Transistor	PJF	p-Kanal Sperrschicht-FET
D	Diode	NMOS	n-Kanal MOSFET
		PMOS	p-Kanal MOSFET

Den **Werten** können wahlfreien Parameter für jedes Halbleiter-Schaltelement zugeordnet werden. Wenn Werte nicht angegeben werden, so werden die DEFAULT-Werte (Ersatzwerte) angenommen.

Bemerkung: Ein definiertes Modell wird in einer Netzliste nur **einmal** angegeben. Auf dieses Modell kann beliebig oft zugegriffen werden.

Jede **Elementzeile** eines Halbleiter-Schaltelements enthält folgende Angaben:

- den Elementnamen
- die Nummern der Schaltungsknoten, mit denen das Element elektrisch verbunden ist
- den Namen des Halbleitermodells.

Zusätzlich können wahlfreie Parameter für jedes Halbleiter-Schaltelement eingegeben werden: Faktoren zu geometrischen Daten und zwei verschiedene Arten von Anfangsbedingungen.

Die Eingabe des Parameters **OFF** setzt eine dieser Anfangsbedingungen; diese dient in der Regel zur Verbesserung der Konvergenz bei Arbeitspunktberechnungen. Dazu werden die Anschlüsse des Halbleiterelements zunächst auf das Potential Null gelegt (Potential des Bezugsknotens). Wenn die iterative Gleichungslösung konvergiert, werden die festgehaltenen Potentiale wieder freigegeben, so daß die Arbeitspunktberechnung nach weiteren Iterationen zum richtigen

Ergebnis führen.

Hat eine Schaltung mehr als einen stabilen Arbeitspunkt, so läßt sich mit diesem Mittel evtl. der gewünschte Schaltungszustand erreichen.

Es ist üblich und einfacher, eine derartige Beeinflussung der Konvergenz bei der Arbeitspunktberechnung mit Hilfe der **NODESET**-Anweisung herbeizuführen.

Format: **.NODESET V(n1)=val1 ... V(nj)=valj**

Beispiel: .NODESET V(1)=2.5V V(7)=120V

Diese Anweisung veranlaßt, daß im **ersten** Rechenzyklus die spezifizierten Knotenspannungen festgehalten werden (vgl. **.IC** Steuerkommando). Die Ergebnisse liefern den Startwert für den zweiten Rechenzyklus, bei welchem die fixierten Knoten wieder freigegeben werden.

Einen Einfluß auf das Ergebnis kann die **.NODESET**-Anweisung nur dann ausüben, wenn *nicht nur ein stabiler Gleichstromzustand* der Schaltung existiert (z.B. Flipflops).

Bei einer Transient-Analyse wird die **.NODESET**-Anweisung ignoriert, falls die Eingabe auch eine **.IC**-Anweisung enthält.


Im Falle beliebig vorgeschriebener Anfangsspannungen der Kondensatoren bzw. Anfangsströme der Spulen empfiehlt sich die direkte Eingabe dieser Werte über die Elementzeilen. Die Arbeitspunktberechnung muß dann durch die **UIC**-Option in der **.TRAN**-Anweisung unterdrückt werden. Alle energiebestimmenden Spannungen und Ströme ohne Anfangswertvorgabe erhalten dabei den Ersatzwert Null.

Der zweite Art von Anfangsbedingungen definiert die Werte der Kapazitätsspannungen zum Zeitpunkt $t = 0$, falls diese bei der Einschwinganalyse vorgegeben werden sollen. Das entsprechende Feld der Elementzeile hat grundsätzlich die Form **IC** = spannungswert.

3.1 Diode

D

Diode




Format: **D**name **NA NC modellname** [**area**] [**OFF**] [**IC=ud(0)**]

Beispiel:

Drect	1	2	DPWR
D1	4	5	N4114 OFF

Der Name muß mit dem Buchstaben **D** beginnen. **NA** ist der Anodenanschluß und **NC** der Kathodenanschluß. Der wahlfreie "**area**"-Faktor gibt die Zahl der äquivalenten parallelgeschalteten Halbleiterelementen an (bezogen auf das Halbleitermodell). Diese Angaben erlauben die Verwendung eines einzigen Halbleitermodells für verschiedene Dioden, die sich lediglich in der Fläche voneinander unterscheiden. Wird "**area**" nicht angegeben, so erfolgt die Standarteinstellung **1**.


Modellparameter der Diode



Name	Bezeichnung	Einheit	Default
IS	(Sperrschicht-)Sättigungsstrom	A	1E-14
RS	Bahnwiderstand	OHM	0
N	Emissionskoeffizient	-	1
TT	Transitzeit (Minoritätsträgerlebensdauer)	sec	0
CJO	Sperrschichtkapazität (bei 0V Vorspannung)	F	0
VJ	Diffusionsspannung (bei 0V Vorspannung)	V	1
M	Gradationsexponent (Sperrschichtkapazität)	-	0.5
EG	Bandabstands-Spannung	eV	1.11
XTI	IS – Temperaturexponent	-	3
KF	Funkelrauschkoeffizient	-	0
AF	Funkelrauschexponent	-	1
FC	Koeffizient für Sperrschichtkapazität im Durchlaßbereich	-	0.5
BV	Durchbruchspannung (rückwärts)	V	∞
IBV	Strom bei der Durchbruchspannung (rückwärts)	I	1E-10

3.2 Bipolartransistor

Q Bipolar Transistor




Format: **Qname NC NB NE [NS] modellname [area] [OFF]**
+ [IC=ube(0), uce(0)]

Beispiel: Q4 4 3 0 QN2222
 Q19 10 32 11 MOD1 IC=0.65, 7

Der Name muß mit dem Buchstaben **Q** beginnen. **NC** ist der Kollektoranschluß, **NB** der Basisanschluß und **NE** der Emitteranschluß. Der wahlfreie "AREA"-Faktor gibt die Zahl der äquivalenten parallelgeschalteten Halbleiterelemente an (bezogen auf das Halbleitermodell). Diese Angaben erlauben die Verwendung eines einzigen Halbleitermodells für verschiedene Dioden, die sich lediglich in der Fläche voneinander unterscheiden. Wird "AREA" nicht angegeben, so erfolgt die Standarteinstellung 1.

NS, der Substratanschluß, ist optional.

Modellparameter des Bipolartransistors



Name	Bezeichnung	Einheit	Default
IS	Transport-Sättigungsstrom	A	1E-16
BF	(Ideale max.) Stromverstärkung (vorwärts)	-	100
NF	Emissionskoeffizient des Vorwärtsstromes	-	1
VAF	Early-Spannung der BC-Diode (vorwärts)	V	∞
IKF	Kniestrom (vorwärts) Grenzwert der Abnahme der Stromverstärkung bei großem Kollektrom	A	∞
ISE	(BE-Leck) Sättigungsstrom der nichtidealen Diode D_{E2}	A	0
NE	(BE-Leck) Emissionskoeffizient der nichtidealen Diode D_{E2}	-	1,5
BR	(Ideale max.) Stromverstärkung (rückwärts)	-	1
NR	Emissionskoeffizient des Rückwärtsstromes	V	1
VAR	Early-Spannung der BE-Diode (rückwärts)	V	∞
IKR	Kniestrom (rückwärts) Grenzwert für die Abnahme der Stromverstärkung bei großem Kollektorstrom	A	∞
ISC	(BC-Leck) Sättigungsstrom der nichtidealen Diode D_{C2}	A	0
NC	(BC-Leck) Emissionskoeffizient der nichtidealen Diode D_{C2}	-	2
RB	Basisbahnwiderstand für kleinen Basistrom	OHM	0


Modellparameter des Bipolartransistors (Fortsetzung)

Name	Bezeichnung	Einheit	Default
IRB	Basisstrom, bei dem $r_{BB}' = 0,5(r_B + r_{BM})$	A	∞
RBM	(Min.) Basisbahnwiderstand für großen Basistrom	OHM	0
RE	Emitterbahnwiderstand	OHM	0
RC	Kollektorbahnwiderstand	OHM	0
CJE	BE-Sperrschichtkapazität bei 0V Vorspannung	F	0
VJE	Diffusionsspannung der BE-Diode bei 0V Vorsprung Vorspannung	V	0.75
MJE	Gradationsexponent der BE-Sperrschichtkapazität	-	0.33
TF	Transitzeit (vorwärts) für kleinen Kollektorstrom	sec	0
XTF	Modellparameter zur Modellierung der Arbeitspunktabhängigkeit (IC) von TF	-	0
VTF	Modellparameter zur Modellierung der Arbeitspunktabhängigkeit (VC) von TF	V	∞
ITF	Modellparameter zur Modellierung der Arbeitspunktabhängigkeit (Hochstrom) von TF	I	0
PTF	zusätzliche Phasendrehung in Grad für $f = 1/2\pi ITF$ der Steilheit	Grad	0
CJC	BC-Sperrschichtkapazität bei 0V Vorspannung	F	0
VJC	Diffusionsspannung der BC-Diode bei 0V Vorspannung	V	0.75
MJC	Gradationsexponent der BC-Sperrschichtkapazität	-	0.33
XCJC	Aufspaltung der Kapazität C_{jC}	-	1
TR	(Ideale) Transitzeit (rückwärts)	sec	0
CJS	Kollektor-Substrat- Sperrschichtkapazität bei 0V Vorspannung	F	0
VJS	Sperrschichtpotential der CS-Diode bei 0V Vorspannung	V	0.75
MJS	Gradationsexponent der CS-Sperrschichtkapazität	-	0
XTB	vorwärts und rückwärts beta Temperaturkoeffizient	-	0

Modellparameter des Bipolartransistors (Fortsetzung)

Name	Bezeichnung	Einheit	Default
EG	Bandabstadsspannung (bandgap voltage)	eV	1.11
XTI	IS - Temperaturexponent	-	3
KF	Funkelrauschkoeffizient	-	0
AF	Funkelrauschexponent	-	1
FC	Koeffizient für Sperrschichtkapazität im Durchlaßbereich (für C_{je} und C_{jc})	-	0.5

3.3 Sperrschicht Feldeffekttransistor

J Junction - FET

Format: **Jname ND NG NS modellname [area] [OFF]**
 + [**IC=uds(0), ugs(0)**]

Beispiel: J1 4 3 0 MOD2

Der Name muß mit dem Buchstaben **J** beginnen. **ND**, **NG** und **NS** sind die Anschlüsse für Drain, Gate und Source. Der wahlfreie "AREA"-Faktor gibt die Zahl der äquivalenten parallelgeschalteten Halbleiterelementen an (bezogen auf das Halbleitermodell). Diese Angaben erlauben die Verwendung eines einzigen Halbleitermodells für verschiedene Dioden, die sich lediglich in der Fläche voneinander unterscheiden. Wird "AREA" nicht angegeben, so erfolgt die Standarteinstellung 1.



Modellparameter des Sperrschicht Feldeffekttransistors

Name	Bezeichnung	Einheit	Default
VTO	Abschnüspannung (Schwellenspannung)	V	-2
BETA	Übertragungsleitwertparameter	A/V ²	1E-4
Lambda	Parameter der Kanallängenmodulation	1/V	0
RD	Drain-Bahnwiderstand	OHM	0
RS	Source-Bahnwiderstand	OHM	0
CGS	Gate-Source-Sperrschichtkapazität bei $U_{GS}=0V$	F	0
CGD	Gate-Drain-Sperrschichtkapazität bei $U_{GD}=0V$	F	0
PB	Gate-Sperrschichtpotential (Diffusionsspannung)	V	1
IS	Gate-Sperr-Sättigungsstrom	I	1E-14
KF	Funkelrauschkoeffizient	-	0
AF	Funkelrauschexponent	-	1
FC	Koeffizient für nichtideale Sperrschichtkapazität im Durchlaßbereich	-	0.5

3.4 MOSFET

M

MOSFET



Format:

Mname ND NG NS NB modellname

+ [L=lwert] [W=wwert] [AD=adwert] [AS=aswert]

+ [PD=pdwert] [PS=pswert] [NRD=nrdwert]

+ [NRS=nrswert] [OFF]

+ [IC=uds(0), ugs(0), ubs(0)]

Beispiel:

M1 4 3 2 0 MOD3 L=120U W06U

Der Name muß mit dem Buchstaben **M** beginnen. **ND**, **NG**, **NS** und **NB** sind die Anschlüsse für Drain, Gate, Source und Body (Bulk).

Die folgende Tabelle erklärt die Bedeutung der verschiedenen wahlfreien Parameter und enthält auch die Ersatzwerte; diese können zum Teil in der **OPTIONS**-Anweisung verändert werden.

Parameter	Bedeutung	Ersatzwert	Bemerkung
lwert	Kanallänge in Meter	100 μ	*
wwert	Kanalbreite in Meter	100 μ	*
adwert	Drain-Diffusionsfläche in Quadratmeter	0	*
aswert	Source-Diffusionsfläche in Quadratmeter	0	*
pdwert	Umfang der Drain-Diffusionsfläche in Meter	0	
pswert	Umfang der Source-Diffusionsfläche in Meter	0	
nrdwert	Faktor zur Berechnung der parasitären Reihenwiderstände RD und RS von Drain	1	
nrswert	Faktor zur Berechnung der parasitären Reihenwiderstände RD und RS von Source	1	

* Ersatzwerte dieser Parameter können durch das OPTIONS-Kommando verändert werden.

Werden bestimmte Modellparameter nicht eingegeben, so ermittelt SPICE diese auf anderem Wege, sofern entsprechende Angaben in der Elementzeile zur Verfügung gestellt werden.

Die folgende Tabelle gibt einen knappen Überblick über die Zusammenhänge.

Parameter der Elementzeile	Beeinflusste Modellparameter
n_{RD}	Sofern keine Angabe zu R_D vorliegt wird $R_D = n_{RD} R_{SH}$ gesetzt.
n_{RS}	Sofern keine Angabe zu R_S vorliegt wird $R_S = n_{RS} R_{SH}$ gesetzt.
A_D A_S	Wenn der Substrat-Sättigungsstrom I_S nicht vorgegeben wird, gilt $I_S = A_D J_S$ für die Drain-Substrat-Sperrschicht und $I_S = A_S J_S$ für die Source-Substrat-Sperrschicht.
A_D, P_D	Berechnung von C_{BD} , falls keine Angabe zum Modellparameter C_{BDO} gemacht wurde.
A_S, P_S	Berechnung von C_{BS} , falls keine Angabe zum Modellparameter C_{BSO} gemacht wurde.

SPICE 2G6 enthält 3 Modelle

LEVEL=1 Shichman-Hodges ("First Order Model")

LEVEL=2 erweitertes physikalisches Modell

LEVEL=3 halbempirisches Modell (weniger Rechenzeit!)

Modellparameter des MOSFET's

Name	Bezeichnung	Einheit	Default
LEVEL	Art des Simulationsmodells (=1,2 und 3)		1
VTO	extrapolierte (Null-) Schwellenspannung	V	0
KP	innerer (Übertragungsleitwert-) Transduktanzparameter	A/V	2E-5
GAMMA	Substrat-Schwellenspannungsparameter	\sqrt{V}	0
PHI	Oberflächenpotential bei starker Inversion	V	0.6
LAMBDA	Parameter der Kanallängenmodulation (LEVEL=1 und LEVEL=2)	1/V	0
RD	Drain-Bahnwiderstand	OHM	0
RS	Source-Bahnwiderstand	OHM	0
CBD	Null Bulk-Drain-Sperrschichtkapazität (bei $U_{BD}'=0V$)	F	0
CBS	Null Bulk-Source-Sperrschichtkapazität (bei $U_{BS}'=0V$)	F	0
IS	Substrat-Sperrättigungsstrom	A	1E-14
PB	Substrat-Sperrschicht-Diffusionsspannung	V	0.8
CGSO	Gate-Source-Überlappungskapazität (pro Meter Kanalbreite)	F/m	0
CGDO	Gate-Drain-Überlappungskapazität (pro Meter Kanalbreite)	F/m	0
CGBO	Gate-Bulk-Überlappungskapazität (pro Meter Kanallänge)	F/m	0
RSH	Diffusionsflächenwiderstand von Drain- und Source	Ohm	0
CJ	Null-Substratbodenkapazität pro Sperrschichtfläche	F/m	0
MJ	Substratboden-Sperrschicht-Gradationskoeffizient	-	0.5
CJSW	Null-Substrat-Seitenwandkapazität pro Meter Sperrschichtumfang (Drain- bzw. Sourceumfang)	F/m	0
MJSW	Substratseitenwand-Sperrschicht-Gradationskoeffizient	-	0.33

Modellparameter des MOSFET's (Forsetzung)

Name	Bezeichnung	Einheit	Default
JS	Substrat - Sperrsättigungsstromdichte	A/m ²	1E-8
TOX	Oxiddicke	M	1E-7
NSUB	Substratdotierungsdichte	1/cm ³	0
NSS	Oberflächenladungsdichte zur Bestimmung von U _{TO}	1/cm ²	0
NFS	Effektive Fast-Surface-Zustandsdichte	1/cm ²	0
TPG	Type des Gatematerials 1 Gegenteil vom Substrate -1 wie Substrate 0 Al Gate	-	1
XJ	metallurgische Sperrschichttiefe	M	1U
LD	Querdiffusionskoeffizient (Gate-Diffusionsüberlappung)	M	0.8U
UO	Oberflächenbeweglichkeit bei niedriger Gate-Spannung	cm ² /Vs	600
UCIRT	kritische Feldstärke für Beweglichkeitsverminderung (LEVEL=2)	V/cm	1E4
UEXP	Exponent der kritischen Feldstärke für Abnahme der Oberflächenbeweglichkeit (LEVEL=2)	-	0
UTRA	Koeffizient der Quersfeldstärke für empirische Beweglichkeitsabnahmeformel (LEVEL=2)	-	0
VMAX	maximale Driftgeschwindigkeit der Ladungsträger	m/s	0
NEFF	Gesamt-Kanalladungskoeffizient (LEVEL=2)	-	1
XQC	Drain-Kanalladungs-Koeffizient	-	1
KF	Funkelrauschkoeffizient	-	0
AF	Funkelrauschexponent	-	1
FC	Koeffizient für Durchlaßbereich der Sperrschichtkapazität	-	0.5
DELTA	Breitenkoeffizient (Korrekturfaktor für Zunahme der Schwellenspannung beim Schmalkanalbaustein) (LEVEL=3)	-	0
THETA	Beweglichkeits-Modulations Faktor (LEVEL=3)	1/V	0
ETA	Statische Rückkopplungsparameter (LEVEL=3)	-	0
KAPPA	Abhängigkeitskoeffizient von der Drain-Feldsättigung (LEVEL=3)	-	0.2

4 Analog Behavioral Modeling (ABM) bei PSpice

Mit der Option "Analog Behavioral Modeling" bei dem Simulator **PSpice** steht dem Schaltungsentwickler ein vielseitiges Werkzeug zur Verfügung. Die Option ABM ist als Erweiterung der beiden gesteuerten Quellen **E** und **G**. Bei SPICE2G6 ist es zwar möglich, nichtlineare gesteuerte Quellen durch mehrdimensionale Polynome zu realisieren, jedoch diese Möglichkeit hat einige wesentliche Einschränkungen:

- Viele Funktionen können durch ein Polynom nicht korrekt beschrieben werden.
- Die Syntax der Polynomspezifikation ist recht umständlich
- Frequenzabhängiges Verhalten lässt sich nicht realisieren.

Mit Hilfe der ABM-Eigenschaft von PSpice werden diese Einschränkungen weitgehend beseitigt. (Bei **LTPSpice** werden die *E* bzw. *G* - ABM Quellen automatisch nach der SPice3 Syntax "B" - Quelle umgesetzt).

4.1 Funktionen (Schlüsselwort: VALUE)

Funktionen können mit dem Schlüsselwort **VALUE** definiert werden:

E<name> N+ N- VALUE = {expression}

G<name> N+ N- VALUE = {expression}

"E" liefert eine Spannung, "G" einen Strom. Der Funktionsausdruck "expression" nach VALUE darf *Spannungen*, *Ströme* und die *Zeit* enthalten. *Ströme müssen* - wie bei SPICE2G6 - als Strom *durch eine Spannungsquelle* (mit dem Spannungswert von 0 V) definiert werden.

Beispiel:

```

Eout 10 0 VALUE = {V(2,3)*V(5,6)}
E_AM out 0 VALUE =
+ {5V*(1+V(3,4))*cos(2*3.14159*460kHz*TIME)}
```

4.2 Look-up Tabelle (Schlüsselwort: TABLE)

E<name> N+ N- TABLE{expression} = (input value, output value)*

G<name> N+ N- TABLE{expression} = (input value, output value)*

Die Übertragungskennlinie wird durch diskrete Wertepaare dargestellt. Zwischen den Werten findet eine lineare Interpolation statt.

Beispiel: idealisierte Operationsverstärker mit der Verstärkung 100 dB

```

EOPid out 0 TABLE{V(in+,in-)} = (-100u,10) (100u,10)
```

4.3 Laplace Transformation (Schlüsselwort: LAPLACE)

E<name> N+ N- LAPLACE {expression} = {transform}

G<name> N+ N- LAPLACE {expression} = {transform}

Beispiel: Tiefpass 1. Ordnung, Zeitkonstante 10ms

```
ETP out 0 LAPLACE {V(in)} = {1/(1+0.01*s)}
```

5 Nichtlinere gesteuerte Quellen bei SPICE3

Format: **Bname N+ N- I = expr**
 Bname N+ N- V = expr

Beispiel:

BIc 4 5 I = Is*exp(Ube/UT)

oder

Bout 10 0 V = V(2,3)*V(5,6)

6 Die Graphische Ausgabe bei PSpice

Neben der Ergebnisausgabe (***.out**) in ASCII Dateien (**.PRINT** - Anweisung) ist die graphische Darstellung der Ergebnisse für die Interpretation der Rechnerergebnisse sehr wichtig. Mit dem graphischen Postprozessor **PROBE** lassen sich in **PSpice** alle berechneten Grössen darstellen. Hierzu wird in die Netzliste (Schaltungsbeschreibung, "**Circuit-File**", ***.cir**) die Anweisung **.PROBE** eingefügt. Dadurch werden standardmäßig alle Knotenspannungen und Ströme in einer Datei (***.dat**) binär abgespeichert.

Nach durchgeführter Simulation kann der Postprozessor **Probe** mit dem Kommando: **Run Probe** zur graphischen Ergebnisauswertung aufgerufen werden.

Mit dem Probe-Kommando **Trace** -> **Add Trace** wird der **gewünschte** Strom- bzw. Spannungsverlauf für einen aufgelisteten Knoten ausgewählt bzw. eingegeben.

Mathematische Beziehungen sind ebenfalls erlaubt und können direkt in die **Add trace** - Kommandozeile eingetippt werden.

Mit dem Menükommando **Plot** -> **Axis Settings** -> **X axis** kann beispielsweise sowohl der Darstellungsbereich (Data Range) als auch die Skalierungsart (**lin** oder **log**) eingestellt werden. Die Befehlsfolge: **Plot** -> **Axis Settings** -> **X axis** -> **Axis Variable** ermöglicht die Wahl einer anderen Knotenspannung bzw. eines anderen Stromes als X-Axenvariable.

Eine zweite Diagrammfläche im Probefenster kann mit der Menüanweisung:

Plot -> **Add plot to Window**.

Beispiel: Bode Diagramm (mit k1, k2 Knotennummer)

im oberen Diagramm: Ausgangsspannung bzw. Verstärkung in [dB] mit dem

Add trace Kommando **VdB(k1,k2)** oder **dB(V(k1,k2))**;

im unteren Diagramm: Phasenverschiebung zwischen Ein- und Ausgangsspannung in [°] mit dem

Add trace Kommando **Vp(k1,k2)** oder **P(V(k1,k2))**

Einige Probe - Trace - Anweisungen:	
Anweisung	Diagrammdarstellung
d(y)	Differenziert die Größe y nach der Laufvariablen x.
dB(y)	Darstellung der auf 1 bezogene normierte Größe y in [dB].
I(x)	Stromfluß durch das Netzwerkelement x.
Img(y)	Darstellung des Imaginärteils einer komplexen Größe y.
P(y)	Phasenwinkel in [°] einer komplexen Größe y.
R(y)	Darstellung des Realteils einer komplexen Größ y.
V(x)	Spannung am Knoten x mit Bezug Knoten 0.
V(x,y)	Spannung zwischen Knoten x und Knoten y.
VdB(x)	Spannung am Knoten x mit Bezug Knoten 0 in [dB].
VdB(x,y)	Spannung zwischen Knoten x und Knoten y in [dB].
Vp(x)	Phasenwinkel in [°] zwischen Spannung am Knoten x und der Eingangsspannung.